Ruban de Moëbius en bandes de cylindres

Robert Ferréol m'a lancé sur le sujet avec  : " … J'ai aussi pensé faire un ruban avec trois cylindres raccordés par des portions "rectilignes" (comme des rotatives d'imprimerie) …" et m'a fait parvenir des photocopies de trois pages d'un ouvrage en allemand ; je ne comprends pas du tout l'allemand mais il y avait heureusement deux schémas très explicites reproduits ci-dessous qui permettent de reconstituer le mécanisme.

J'ai d'abord reproduit le ruban ci-dessus  à 3 cylindres puis généralisé à n cylindres en suivant le même principe exposé ci-dessous.

Pour trouver la valeur de x = r (n  2)/2, j'ai résolu l'équation exprimant la colinéarité du segment [b0, a1] et dir( /n ). (Voir après pour les notations utilisées dans cette ligne )

On considère n entier fixé, d, r des réels strictement positifs.

On pose a = 2/n et x = r (n  2)/2.
et la fonction dir :      [cos(), sin(), 0]

 Soient :

1. l'arc d'hélice d'équation HEL := [​​​  r cotan(a/2), d + r cos(t),  r sin(t)] , t  [ /2, /2] ,  tracée sur le cylindre : (u,t)    cyl =  [u, d + r cos(t),  r sin(t)] .

2. l'arc d'hélice hel0 := [ (n  1) r cotan(a/2) t, d+(n  1) r cos(t), (n  1) r sin(t)] , t  [ /2, /2] ,  tracée sur le cylindre : (u,t)    cyl =  [u, d + (n  1) r cos(t), (n  1) r sin(t)] .

La tangente en un point de ces arcs fait un angle de /n avec la génératrice passant par ce point.

Pour i= 1 à  n, on désigne par heli l'arc d'hélice déduite de HEL par une rotation d'angle i.a d'axe Oz suivie de la translation de vecteur  V : V = x dir(ia + /2) + [0, 0, (n  2 i) r] dont l'équation est donc :
 heli := [r  sin(i a) -  sin(i a) r  n/2 -  r t cotan(a/2) cos(i a) - d sin(i a) - r sin(i a) cos(t),

        - r  cos(i a) + r  n cos(i a) /2 - r t cotan(a/2) sin(i a) + d cos(i a) + r cos(i a) cos(t),    r n - 2 r i - r sin(t)]

les extrémités sont les points : ai obtenu pour t = /2 et bi pour t = /2.
a0 et b0 sont les extrémités de hel0. 

Les segments raci = [bi , ai], i= 1..n  1,  ( et rac0 = [b n , a0] ) sont colinéaires aux tangentes en bi  et  ai aux arcs d'hélices d'extrémités ces points.

La courbe réunion des heli et raci, i = 0 à n  1 sans points anguleux.

Soit L une longueur fixée.

Soient les points a'i et a"i déduits de ai par les translations  L dir(ia)  et  –L dir(ia)  et de même pour les bi.

Les polygones [b'i , b"i , a"i , a'i] sont plans et se raccordent sans points anguleux  aux bandes de cylindres Cyli :

(u,t)   heli + u dir(ia) pour  u  [-L, L] et t   [ /2, /2] .

Le ruban ainsi constitué présente n  2 torsions d'un demi-tour et est développable en une bande de largeur constante à bords rectilignes.

Ci-dessous le programme Maple version 7 ou 8 ( pour les versions 4 ou 5, il faut remplacer les nombreuses doubles barres "||" par des points "." )

# Alain Esculier – juin 2003 -ruban développable à  (n-2) demi-tours

 restart: with(plots):

 rot:=(M,u) -> [M[1]*cos(u)-M[2]*sin(u),M[1]*sin(u)+M[2]*cos(u),M[3]]:

cyl:= (dir,large) -> expand(large+u*dir):

cof:=2*r*cos(t0/2)/sin(t0/2)*Pi/2+large*1.2:

dir:=t ->[cos(t),sin(t),0]:

he2:= [-r*cos(t0/2)/sin(t0/2)*t,d+r*cos(t),-r*sin(t)]:

he0:= [-r*cos(t0/2)/sin(t0/2)*t,d+r*cos(t),r*sin(t)]:

cylbase:=[u,d+(r-0.1)*cos(t),(r-0.1)*sin(t)]:

deb:=-Pi/2: fin:=Pi/2: rge_t:=deb..fin:

# --------------------------------------

hel||0:=subs(r=(n-1)*r,he0):

a0:=simplify(subs(t=deb,hel0)): b0:=simplify(subs(t=fin,hel0)):

# ------------- ces valeurs peuvent être changées ----------------------

n:=4:  # ****** n > 2 

d:=8:   # **** si n grand il faut augmenter d en conséquence ***

large:=5: # **** fixe la largeur de la bande   **** 

 r:=1: 

colC:=red:

# ------------------------------------

x:=-r*Pi+1/2*r*Pi*n:
t0:=+2*Pi/n:

for i from 1 to n-1 do 

   hel||i:=expand(rot(he2,i*t0)+x*dir(i*t0+Pi/2)+[0,0,(n-2*i)*r]):

   a||i:=simplify(subs(t=deb,hel||i)); b||i:=simplify(subs(t=fin,hel||i));

   rac||i:=[b||(i-1),a||i];

   cyl||i:=expand(rot(cylbase,i*t0)+x*dir(i*t0+Pi/2)+[0,0,(n-2*i)*r]):

od:

rac||0:=[b||(n-1),a||0]:

cyl0:=[u,d+((n-1)*r-0.1)*cos(t),((n-1)*r-0.1)*sin(t)]: 

for i from 0 to n-1 do

    H||i:=spacecurve(hel||i,t=rge_t,color=red,thickness=2): 

    Rac||i:=spacecurve(rac||i,color=red,thickness=2):

    Rub||i:=plot3d(expand(hel||i + u*dir(i*t0)), u=-large..large,t=rge_t,grid=[2,50]);

    Rubrac||i:=polygonplot3d([ b||i-large*dir(i*t0), b||i+large*dir(i*t0),

                            a||((i+1) mod n)-large*dir(((i+1) mod n)*t0),

                              a||((i+1) mod n)+large*dir(((i+1) mod n)*t0)]

                            , style=patchnogrid):

# ********** mettre patch pour voir pb si n grand et d trop petit

    CYL||i:=plot3d(cyl||i,u=-cof..cof,t=0..2*Pi,grid=[2,50],style=line,color=grey):

od:

surf:=display([seq(op([ Rub||i,Rubrac||i ]), i=0..n-1)],style=patchnogrid):

courbe:=display([seq(op([ H||i,Rac||i ]), i=0..n-1)],color=red,thickness=2):

rouage:=display([seq(CYL||i, i=0..n-1)],style=line):
plotsetup(window): #plotsetup(inline):

display([surf,courbe,rouage ],title=cat(n," cylindres et ",n-2, " demi-tours"),scaling=constrained,orientation=[10,70],lightmodel=light2);#

